

Tight-Binding 法の基礎

東京大学大学院工学系研究科機械工学専攻 泉聡志

2003-11-05

Tight-Binding 法 (TB 法) は量子効果をポテンシャル計算に取り込む最もシンプルなアプローチである。結合の方向性が自然な形で取り込まれる。また、結合・反結合状態の考慮ができることにより、結合の記述が高精度化する。

1 直交二中心 TB 近似

TB 法では、分子軌道法と同じく、 n 番目の固有状態の波動関数を式 (1) のように原子軌道の線形結合で表す。ここで、 $\phi_{i\alpha}$ は原子 i の軌道 α を示す。

$$\psi^{(n)} = \sum_{i\alpha}^N C_{i\alpha}^{(n)} \phi_{i\alpha} \quad (1)$$

これを Kohn-Sham 方程式 $\mathcal{H}\psi^{(n)} = \varepsilon^{(n)}\psi^{(n)}$ に代入し¹、両辺に $\phi_{i\alpha}$ を左からかけると式 (2) が得られる。

$$\sum_j^N \mathcal{H}_{i\alpha,j\beta} C_{j\beta}^{(n)} = \varepsilon^{(n)} \sum_{j\beta} S_{i\alpha,j\beta} C_{j\beta}^{(n)} \quad (2)$$

$$(H_{i\alpha,j\beta} = \int \phi_{i\alpha}^* \hat{H} \phi_{j\beta} dV) \quad (3)$$

$$(S_{i\alpha,j\beta} = \int \phi_{i\alpha}^* \phi_{j\beta} dV) \quad (4)$$

ここで、次のような近似を行う。

1. 最小限の基底を扱う。

角運動量・磁気量子数それぞれに 1 つの軌道とする。

2. 直交 TB モデル

重なり積分は対角行列とする ($i\alpha \neq j\beta$ の積分を無視する)。直交化 Lowdin 基底関数を使う場合は、自動的に対角行列となるが、非直交化基底関数を使う場合は、重なり積分に非対角項が現れる。非対角項の無視は原子間行列要素のシフトと平均一粒子エネルギーのシフトで解決できる。平均一粒子エネルギーのシフトは反発の 2 体ポテンシャルの項に吸収できる。

3. ハミルトニアンのカットオフ

第三近接程度でカットオフする (bcc などは第三近接以上必要)。

4. 二中心近似

KS 方程式の有効ポテンシャルは

¹ $\mathcal{H} = -\frac{\hbar^2}{2m} \nabla^2 + V_{eff}(\mathbf{r})$

$$V_{eff} = \sum_k V_{eff,k}(\mathbf{r} - \mathbf{r}_k) \quad (5)$$

となるので、ハミルトニアン行列は以下の式となる。

$$H_{i\alpha,j\beta} = \int \phi_{i\alpha}^* \left(-\frac{\hbar^2}{2m} \nabla^2 + \sum_k V_{eff,k} \right) \phi_{j\beta} dV \quad (6)$$

ここで

- 原子 $i \neq j \neq k$ の三中心積分を無視する²。ハミルトニアンポテンシャル項は軌道が位置する i, j の2原子によるポテンシャル関数に置き換えることができる。
- 二中心行列要素は Slater-Koster 要素 $V_{ll'm}$ (同じ磁気量子数の軌道間の行列要素のみに生じる) の線形和で表現できる。

具体的なハミルトニアン行列については2で述べる。

重なり積分を対角行列とした近似より式(7)の N 次永年方程式を得る。この永年方程式は固有値 ε^n 、固有ベクトル $C_{i\alpha}^n$ を求める固有値問題であり、マトリックス H の対角化によって解ける。

$$\sum_j^N \mathcal{H}_{i\alpha,j\beta} C_{j\beta}^{(n)} = \varepsilon^{(n)} C_{i\alpha}^{(n)} \quad (H_{i\alpha,j\beta} = \int \phi_{i\alpha}^* \hat{H} \phi_{j\beta} dV) \quad (7)$$

C (ベクトル) は固有値なので、以下の直交性・規格化・完全性が成り立つ。

$$(C^n)^* C^m = \delta^{nm} \quad (8)$$

$$|C^n|^2 = 1 \quad (9)$$

$$|C^n \rangle \langle C^n| = I \quad (10)$$

パウリの原理により、もし系が $2z$ 個の電子を持っていたなら、小さいほうから z 個の固有値が占有され、バンドエネルギーは

$$E_{band} = 2 \sum_{n=1}^z \varepsilon^{(n)}$$

となる。永年方程式の表現と結合すると、

$$\begin{aligned} E_{band} &= \sum_{i\alpha,j\beta} 2 \sum_{n=1}^z C_{i\alpha}^{(n)} C_{j\beta}^{(n)*} H_{j\beta,i\alpha} \\ &= 2 \sum_{i\alpha,j\beta} \rho_{i\alpha,j\beta} H_{j\beta,i\alpha} (= 2 \text{Tr}[\rho H]) \end{aligned} \quad (11)$$

ここで、密度マトリックスを

$$\rho_{i\alpha,j\beta} = \sum_{n=1}^z C_{i\alpha}^{(n)} C_{j\beta}^{(n)*}$$

で定義した。

²分子・固体の結合と構造 p190

密度マトリックスの対角項は電子数に対応 ($\hat{N}^e = 2\text{Tr}[\rho]$) し、非対角項はボンドオーダー³に対応することが知られている。

また、総電子密度は式 (12) で示される。

$$\rho(\mathbf{r}) = 2 \sum_{n=1}^z |\psi^n(\mathbf{r})|^2 \quad (12)$$

原子間力はヘルマンファイマン則より、

$$\begin{aligned} \frac{\delta E_{band}}{\delta \mathbf{r}} &= \sum_{i\alpha, j\beta} \left(\rho_{i\alpha, j\beta} \frac{\partial \mathcal{H}_{j\beta, i\alpha}}{\partial \mathbf{r}} + \rho_{j\beta, i\alpha} \frac{\partial \mathcal{H}_{i\alpha, j\beta}}{\partial \mathbf{r}} \right) \\ &= -2\text{Tr} \left[\rho \frac{\partial \mathbf{H}}{\partial \mathbf{r}} \right] \end{aligned} \quad (13)$$

2 TB ハミルトニアン行列要素

タイトバインディング計算では、積分を距離と角度に依存したパラメータの式で近似する。第一近似として、ハミルトニアンポテンシャル部分を2つの原子のみによるポテンシャルに置き換え、以下の式を得る。

$$H_{i\alpha, j\beta} = \sum_J h_{\alpha\beta J}(r_{ij}) G_{\alpha\beta J}(k, l, m) \quad (14)$$

J は角運動量 (Angular momentum) である ($0 \rightarrow \sigma, 1 \rightarrow \pi, 2 \rightarrow \delta$)。 $G_{\alpha\beta J}(k, l, m)$ は角度依存項である。角度依存項は Slater-Koster[1] の論文の表 1 に示されている。表 1 にいくつかの抜粋を示す。

$h_{\alpha\beta J}(r)$ は TB パラメータと呼ばれ数々のモデルが提案されている。対称性のため、 $sp_x\sigma, sp_y\sigma, sp_z\sigma$ は同じ h の形にして角度項をかける

表 1: Energy integrals for crystal in terms of two-center integrals ($l = \frac{r_{ijx}}{r_{ij}}, m = \frac{r_{ijy}}{r_{ij}}, n = \frac{r_{ijz}}{r_{ij}}$)

$E_{s,s}$	$h_{ss\sigma}$
$E_{s,x}$	$lh_{sp\sigma}$
$E_{x,x}$	$l^2 h_{pp\sigma} + (1-l^2) h_{pp\pi}$
$E_{x,y}$	$lmh_{pp\sigma} - lmh_{pp\pi}$
$E_{x,z}$	$lnh_{pp\sigma} - lnh_{pp\pi}$
$E_{s,xy}$	$\sqrt{3}lmh_{sd\sigma}$
E_{s,x^2-y^2}	$\frac{1}{2}\sqrt{3}(l^2 - m^2)h_{sd\sigma}$
$E_{s,3z^2-r^2}$	$[n^2 - \frac{1}{2}(l^2 + m^2)]h_{sd\sigma}$
$E_{x,xy}$	

例えば、シリコンの sp 軌道の場合二原子の TB 行列要素は以下のような形になる。

³ボンド間の電子数〜(結合電子数-反結合電子数)

$$\begin{pmatrix}
E_s & 0 & 0 & 0 & h_{ss\sigma} & h_{sp\sigma}l & h_{sp\sigma}m & h_{sp\sigma}n \\
& E_p & 0 & 0 & -h_{sp\sigma}l & h_{pp\sigma}l^2 + h_{pp\pi}(1-l^2) & (h_{pp\sigma} - h_{pp\pi})lm & (h_{pp\sigma} - h_{pp\pi})ln \\
& & E_p & 0 & -h_{sp\sigma}m & (h_{pp\sigma} - h_{pp\pi})lm & h_{pp\sigma}m^2 + h_{pp\pi}(1-m^2) & (h_{pp\sigma} - h_{pp\pi})mn \\
& & & E_p & -h_{sp\sigma}n & (h_{pp\sigma} - h_{pp\pi})ln & (h_{pp\sigma} - h_{pp\pi})mn & h_{pp\sigma}n^2 + h_{pp\pi}(1-n^2) \\
& & & & E_s & 0 & 0 & 0 \\
sym. & & & & & E_p & 0 & 0 \\
& & & & & & E_p & 0 \\
& & & & & & & E_p
\end{pmatrix} \quad (15)$$

TB 永年方程式を解くためには、on-site エネルギー (第一イオン化エネルギー) と原子間行列要素が必要である。例えば、シリコンの場合 s と p の on-site エネルギー E_s, E_p と、s と p 軌道の 4 つの結合 $h_{\alpha\beta J}(ss\sigma, sp\sigma, pp\sigma, pp\pi)$ を設定しなくてはならない。

Chadi は、シリコンの経験的直交化 TB モデルを提案している [2][3]。 $E_s=-5.20\text{eV}$, $E_p=1.20\text{ eV}$, $h_{ss\sigma}(r_0)=-1.94\text{ eV}$, $h_{sp\sigma}(r_0)=1.75\text{ eV}$, $h_{pp\sigma}(r_0)=3.05\text{ eV}$, $h_{pp\pi}(r_0)=-1.08\text{ eV}$, $r_0=2.36\text{ \AA}$ 。

$$\begin{aligned}
h_{ss\sigma} &= h_{ss\sigma}(r_0)r_0^2/r^2, & h_{sp\sigma} &= h_{sp\sigma}(r_0)r_0^2/r^2 \\
h_{pp\sigma} &= h_{pp\sigma}(r_0)r_0^2/r^2, & h_{pp\pi} &= h_{pp\pi}(r_0)r_0^2/r^2
\end{aligned}$$

距離依存の項をスケーリング関数と呼ぶ。反発力は Wolfsberg-Helmholz の式 ($\Phi_{rep} \simeq A[h(R)]^2$) から定義できる。

また、Goodwin[4] らも、異なる配位数に対応する TB モデルを提案している (GSP モデル)。

ボンド積分と反発力は以下の式のような形で定義されている。

$$h_J(r_0/r) = h_J(1)(r_0/r)^n \exp[n(-(r/r_c)^{n_c} + (r_0/r_c)^{n_c})] \quad (16)$$

$$\Phi(r_0/r) = \Phi(1)(r_0/r)^m \exp[m(-(r/r_c)^{n_c} + (r_0/r_c)^{n_c})] \quad (17)$$

ここで $E_s=-13.08\text{ eV}$, $E_p=-4.785\text{ eV}$, $h_{ss\sigma}(1)=-1.82\text{ eV}$, $h_{sp\sigma}(1)=1.96\text{ eV}$, $h_{ps\sigma}(1)=-h_{sp\sigma}(1)$, $h_{pp\sigma}(1)=3.06\text{ eV}$, $h_{pp\pi}(d_0)=-0.87\text{ eV}$, $n=2$, $m=4.54$, $r_c=3.67\text{ \AA}$, $n_c=6.48$, $\Phi(1)=3.4581$ 。

また、Wang らのグループらは炭素 [5]、シリコン [6] の TB ポテンシャルを提案している。これらのポテンシャルはカットオフ距離が 1 種類で結晶構造に依存しないように改良された。また、これらのポテンシャルのプログラムが Colombo[7] によって配布されている。

3 エネルギー

総エネルギーは、エネルギーバンド形成によるエネルギー⁴と、原子核の反発エネルギーの和から、自由原子のエネルギーを引いたものになる。

$$E_{tot} = E_{band} + E_{rep} - E_{atom} \quad (E_{atom} = N_{i\alpha}^{atom} \varepsilon_{i\alpha}) \quad (18)$$

⁴孤立原子が結合を作ることにより、孤立状態より低い結合準位を作り出す、その準位にエネルギーが低い順に電子が占有していくので、トータルのエネルギーは低下する

4 参考（電荷の扱いについて）

TB 法では、ハミルトニアンは電荷分布に依存せず、イオン結合など電荷の移動が著しい場合は破綻してしまう。電荷についてセルフコンシストを実現する方法として、局所電荷中和条件と、Hubbard U と、ab-initioTB 法がある（使われる順）。

4.1 局所電荷中和条件 (LCN)

全体の電荷は、もとの孤立状態と変わらないように、フェルミエネルギーが調節されるが、格子欠陥がある場合などは電子構造が変化して、電子数の過不足が生じる。このように欠陥のまわりでは、電荷の中性が保たれない。そこで、各サイトの電荷を中性に保ち、式 (19) が成立するようにそれぞれの原子軌道の on-site エネルギーの差 ($\varepsilon_p - \varepsilon_s$) を一定にさせたまま、on-site エネルギーを $\Delta\varepsilon_i$ だけシフトさせる ($\varepsilon_{i\alpha} = \varepsilon_{i\alpha}^0 + \Delta\varepsilon_i$)。つまりは、電子数を強制的に調節する。金属結合等でよくもちいられる手法である。

$$\sum_{\alpha} [2\rho_{i\alpha,i\alpha} - N_{i\alpha}^{atom}] = 0 \quad (19)$$

まわりの影響を受けるために、この操作は全原子にわたってセルフコンシストに行う必要がある。これにより、LCN を課した昇位エネルギーは式 (20) となる。

$$U_{prom}^{LCN} = \sum_{i\alpha} [2\rho_{i\alpha,i\alpha} - N_{i\alpha}^{atom}] \varepsilon_{i\alpha}^0 \quad (20)$$

この作業は力には直接的には影響しない。

4.2 Hubbard U

共有結合などで電荷の移動が著しいときは、LCN は使えない。Hubbard U では

$$H_u = \frac{1}{2} \sum_i (Q_i - Z_i)^2 \quad (21)$$

を系の結合エネルギーに加える。ハミルトニアン上では下記の手順となる。

Hubbard U では、ハミルトニアン on-site エネルギーに一つの項 (式 22) を加える。 Q_i が変化しなくなるまで、セルフコンシストに解く必要がある [4][8]。

$$\delta E_i = U \times (Q_i - Z_i) \quad (22)$$

Z_i は原子核の電荷 (シリコンは 4)[9]、 Q_i は原子 i 上の電荷⁵、U は Hubbard の U で 1~20[eV] の間にあり、結果は U に依存しないといわれている。

LCN は U の値を無限大にして、 $Q_i = Z_i$ になるようにした手法である。

⁵ $Q_I = \sum_{k\alpha} |c_{i\alpha}^k|^2 f_k$

5 計算量の問題

TB 法はマトリクスの対角化が必要なため、大きな系への適用（100 個程度が限度だと考えられる）が難しい。TB 法の計算量を減らして、Order-N に近づけるために、数多くの研究がなされている。

6 プログラム解説

6.1 サブルーチンリスト

```
c      subroutine bondlength: compute bond-length distribution function
c      subroutine correlate : compute particle-particle correlation function and
c                           radial distributon function
c      subroutine crystal   : assigne tight-binding and repulsive potential parameters
c      subroutine dosenergy : compute total electronic density of states
c      subroutine feynman   : compute Hellmann-Feynman forces
c      subroutine htb       : compute and diagonalizes the tight-binding matrix;
c                           compute band-structure energy
c      subroutine init_pos  : assigne initial positions
c      subroutine init_vel  : assigne initial velocities
c      subroutine md        : molecular-dynamics loop
c      subroutine meandisp  : compute the atomic mean square displacement
c      subroutine nearangle : compute bond-angle distribution function and
c                           atomic coordination number
c      subroutine repuls    : compute repulsive energy and forces
c      subroutine temper    : compute temperature
```

6.2 サブルーチン MD

```
C*****
      subroutine md(itype)
C*****
c      RELEASE TBMD.2.0
c      Last revision: November 21, 1996
C*****
c      Contact author:   Luciano Colombo
c                        e-mail:  colombo@mi.infn.it
C*****
c
c      Subroutine MD updates atomic positions, velocities and accelerations
c      according to the velocity Verlet algorithm.
c
c      Variables (see also main program for futher information)
c
c      - x/y/z(i)       : position of the i-th atom
c      - vx/y/z(i)      : velocity of the i-th atom
c      - ax/y/z(i)      : acceleration of the i-th atom
c      - ax/y/zold(i)   : acceleration of the i-th atom at the previous step
c      - xx/yy/zz(i,J)  : relative position of the j atom to the i atom
c      - rr2(i,j)       : squared relative distance of j-th to i-th atom
c      - frepx/y/z(i)   : repulsive force for the i-th atom
c                       (reduced units : eV/A= 1.602177e-4 dyne)
c      - fhfx/y/z(i)   : Hellmann-Feynman force for the i-th atom
c                       (reduced units : eV/A= 1.602177e-4 dyne)
```

```

c      - ftotx/y/z(i) : total force for the i-th atom
c                      (reduced units : eV/A= 1.602177e-4 dyne)
c
c*****
c
c      Include files
c
c      include 'param.inc'
c      include 'pos.inc'
c      include 'vel.inc'
c      include 'acc.inc'
c      include 'dist.inc'
c      include 'latt.inc'
c      include 'tb.inc'
c      include 'frep.inc'
c      include 'fhf.inc'
c      include 'ftot.inc'
c      include 'erg.inc'
c      include 'eigen.inc'
c      include 'func.inc'
c      include 'lst.inc'
c      include 'simul.inc'
c      include 'start.inc'
c
c      real*8 ax(n), ay(n), az(n)
c
c      Evolution of the system according to 'VELOCITY VERLET' algorithm
c
c      do 1 i=1,n
c          x(i)=(x(i)+ dt*vx(i) + dtsqr*axold(i))
c          y(i)=(y(i)+ dt*vy(i) + dtsqr*ayold(i))
c          z(i)=(z(i)+ dt*vz(i) + dtsqr*azold(i))
c
c          x(i)=x(i)-cell*(dint(x(i)/cell+1.d0)-1.d0)
c          y(i)=y(i)-cell*(dint(y(i)/cell+1.d0)-1.d0)
c          z(i)=z(i)-cell*(dint(z(i)/cell+1.d0)-1.d0)
1      continue
c
c      do 2 i=1,n
c          do 3 j=1,n
c              xx(j,i)= 0.d0
c              yy(j,i)= 0.d0
c              zz(j,i)= 0.d0
c              rr2(j,i)= 0.d0
3          continue
2      continue
c
c      Calculation of the relative position and of the relative
c      squared distance of any atom to the nearest image of any other
c      one, according to periodic boundary conditions.
c
c      do 4 i=1,n
c          do 5 j=i+1,n
c              xx(i,j)= x(j)-x(i)
c              xx(i,j)=xx(i,j)-cell*(dint((xx(i,j)/cell+3.d0)/2.d0)-1.d0) !周期境界条件 x
c              yy(i,j)= y(j)-y(i)
c              yy(i,j)=yy(i,j)-cell*(dint((yy(i,j)/cell+3.d0)/2.d0)-1.d0) !周期境界条件 y

```

```

zz(i,j)= z(j)-z(i)
zz(i,j)=zz(i,j)-cell*(dint((zz(i,j)/cell12+3.d0)/2.d0)-1.d0) !周期境界条件 z
rr2(i,j)= xx(i,j)**2 + yy(i,j)**2 + zz(i,j)**2 !原子間距離の二乗
c
xx(j,i)=-xx(i,j)
yy(j,i)=-yy(i,j)
zz(j,i)=-zz(i,j)
rr2(j,i)= rr2(i,j)
5  continue
4  continue
c
c  Compute Verlet list ! book-keeping リストの計算
c
do 10 i=1,n
  ilst=0
  jlst=0
  do 20 j=1,n
    if ( i .eq. j ) goto 20
    if ( rr2(i,j) .gt. cutoff ) goto 20
    ilst=ilst+1
    list(i,ilst)=j
    if ( i .gt. j ) goto 20
    jlst=jlst+1
    list1(i,jlst)=j
20  continue
    nlst(i)=ilst
    nlst1(i)=jlst
10  continue
c
c  Compute repulsive forces
c
call repuls(itype) ! 反発二体ポテンシャルの計算
c
c  Compute and diagonalizes tight-binding hamiltonian matrix.
c
call htb ! タイトバインディングハミルトニアン行列の計算とその対角化
c
c  Compute of Hellmann-feynman forces
c
call feynman ! ヘルマン・ファインマン力（原子間力）の計算
c
do 8 i=1,n
c
c  Compute total forces and accelerations
c
  ftotx(i)= frepx(i) + fhfx(i)
  ax(i)= ftotx(i)/mass
c
  ftoty(i)= frepy(i) + fhfy(i)
  ay(i)= ftoty(i)/mass
c
  ftotz(i)= frepz(i) + fhfz(i)
  az(i)= ftotz(i)/mass
c
c  Update velocities
c
  vx(i)= vx(i) + dthalf * (ax(i) + axold(i))

```

```

        vy(i)= vy(i) + dthalf * (ay(i) + ayold(i))
        vz(i)= vz(i) + dthalf * (az(i) + azold(i))
c
c      Store accelerations
c
        axold(i)= ax(i)
        ayold(i)= ay(i)
        azold(i)= az(i)
c
8      continue
c
c      Compute instantaneous temperature
c
        call temper
c
        return
        end

```

6.3 サブルーチン htb

```

subroutine htb
C*****
c      RELEASE TBMD.2.0
c      Last revision: November 21, 1996
C*****
c      Contact author:   L. Colombo
c                      e-mail:  colombo@mi.infn.it
C*****
c
c      Subroutine HTB computes and diagonalizes the tight-binding hamiltonian
c      matrix.
c      LAPACK library for matrix diagonalization is used.
c
c      Optimized use of memory
c
c      The actual dimension of the hamiltonian matrix to diagonalize
c      are set to (n4+1,n4+1). This allows for avoiding bank conflicts
c      The extrac line and column of h is filled by zero matrix entries,
c      but for the (n4+1,n4+1) element: it is set equal to 100.d0,
c      i.e. a number well outside the typical energy spectrum of the
c      TB matrix.
c
C*****
c
c      Include files.
c
        include 'param.inc'
        include 'dist.inc'
        include 'eigen.inc'
        include 'latt.inc'
        include 'tb.inc'
        include 'func.inc'
        include 'lst.inc'
        include 'erg.inc'
        include 'potz.inc'
c

```

```

integer lwork,info
parameter(lwork=34*(n4+1))
c
real*8 xxn, yyn, zzn, rr2n, rrn
real*8 scale1, scale2, scale3, scale4
real*8 etot
real*8 work(lwork)
character*1 jobz,uplo
c
do 10 i1=1,n4+1
  do 20 i2=1,i1-1
    h(i2,i1)=0.d0
20  continue
10  continue
h(n4+1,n4+1)=100.d0
c
c  Compute hamiltonian matrix
c
do 100 i=1,n
c
  i43= 4*i-3
  i42= 4*i-2
  i41= 4*i-1
  i40= 4*i
c
  do 200 ji=1,nlst1(i)
    j= list1(i,ji)
c
    j43= 4*j-3
    j42= 4*j-2
    j41= 4*j-1
    j40= 4*j
c
    rr2n= rr2(i,j)
    rrn= dsqrt(rr2n)
c
    xxn= xx(i,j)/rrn
    yyn= yy(i,j)/rrn
    zzn= zz(i,j)/rrn
c
c  Scaling functions for hopping integrals
c  ハミルトニアン行列要素のためのスケーリング関数 (距離依存)
c  scale1= (r0/rrn)**2 * ! h_{ss \sigma}
*      dexp( 2.d0 *(-(rrn/rc1)**enc1 + (r0/rc1)**enc1))
c  scale2= (r0/rrn)**2 * ! h_{sp \sigma}
*      dexp( 2.d0 *(-(rrn/rc2)**enc2 + (r0/rc2)**enc2))
c  scale3= (r0/rrn)**2 * ! h_{pp \sigma}
*      dexp( 2.d0 *(-(rrn/rc3)**enc3 + (r0/rc3)**enc3))
c  scale4= (r0/rrn)**2 * ! h_{pp \pi}
*      dexp( 2.d0 *(-(rrn/rc4)**enc4 + (r0/rc4)**enc4))
c
  h(i43,j43)= sss * scale1 ! h_{ss \sigma}
  h(i43,j42)= sps * xxn * scale2 ! h_{sp \sigma} r_{ijx}/r_{ij}
  h(i43,j41)= sps * yyn * scale2 ! h_{sp \sigma} r_{ijy}/r_{ij}
  h(i43,j40)= sps * zzn * scale2 ! h_{sp \sigma} r_{ijz}/r_{ij}
c
  h(i42,j43)= - h(i43,j42)

```

```

c h_{pp \sigma} r_{ijx}^2/r_{ij}^2 + h_{pp \pi}*(1-r_{ijx}^2/r_{ij}^2)
  h(i42,j42)= ( pps * scale3 * xxn**2 +
    * ppp * scale4 * ( 1.d0 - xxn**2 ) )
c (h_{pp \sigma} - h_{pp \pi}) r_{ijx}/r_{ij} r_{ijy}/r_{ij}
  h(i42,j41)= ( pps * scale3 - ppp * scale4 ) * xxn * yyn
  h(i42,j40)= ( pps * scale3 - ppp * scale4 ) * xxn * zzn
c
  h(i41,j43)= - h(i43,j41)
  h(i41,j42)= h(i42,j41)
c
  h(i41,j41)= ( pps * scale3 * yyn**2 +
    * ppp * scale4 * ( 1.d0 - yyn**2 ) )
  h(i41,j40)= ( pps * scale3 - ppp * scale4 ) * yyn * zzn
c
  h(i40,j43)= - h(i43,j40)
  h(i40,j42)= h(i42,j40)
  h(i40,j41)= h(i41,j40)
c
  h(i40,j40)= ( pps * scale3 * zzn**2 +
    * ppp * scale4 * ( 1.d0 - zzn**2 ) )
c
200 continue
c
  h(i43,i43)= ess ! 対角成分
  h(i42,i42)= epp
  h(i41,i41)= epp
  h(i40,i40)= epp
c
100 continue
c
c Diagonalization
c
  jobz='V'
  uplo='U' ! 対角化サブルーチン (対称マトリックス用)
  call dsyev(jobz,uplo,n4+1,h,n4+1,eval,work,lwork,info)
c h には固有ベクトル、eval には固有値が戻される。
  etot = 0.d0
c
c Compute the statistical weight for any energy level.
c Zero-temperature Fermi-Dirac distribution is assumed.
c
do 31 i=1,n4+1
  if (i.gt.n4/2) then
    occup(i) = 0.d0 ! 半分は空いている
  else
    occup(i) = 2.d0 ! 半分は詰まっている
  end if
31 continue
c
c Compute band-structure energy
c
do 30 i=1,n4+1
  etot = etot + eval(i)*occup(i) ! 低いエネルギー (固有値) より二つずつつめていく
30 continue
c
c Compute band-structure energy per atom
c

```

```

    ebstot = etot / dfloat(n)
c
    return
end

```

参考文献

- [1] J. Slater, G. Koster, Phys. Rev. 94(1954), 1498
- [2] C. Z. Wang, C. T. Chen, K. M. Ho, Phys. Rev. B39 (1989), 8586
- [3] D. J. Chadi, Phys. Rev. Lett. , 41 (1978), 1062; Phys. Rev. B29 (1984), 785
- [4] L. Goodwin, A. J. Skinner, D. G. Pettifor, Europhys. Lett. , 9 (1989), 701
- [5] C. H. Xu, C. Z. Wang, C. T. Chan, K. M. Ho, J. Phys. :Condens. Matter., 4 (1992), 6047
- [6] I. Kown, R. Biswas, C. Wang, K. Ho, C. Soukoulis, Phys. Rev. B 49 (1994), 7242
- [7] L. Colombo, Comp. Mater. Sci., 12 (1998), 278
- [8] M. W. Finnis, A. T. Paxton, D. G. Pettifor, A. P. Sutton, Y. Ohta, Philo. Mag. A, 58 (1988),143
- [9] M. Kohyama, Phys. Rev. B 49(1994), 17102